
IX.5.3  FORECAST SYSTEM PREPROCESSOR DATA BASE ACCESS

Introduction

The Preprocessor Data Base (PPDB) [Hyperlink] contains the observed
data and forecast data used by the Preprocessor Functions to produce
time series.

Access Routines

A description of the routines that can be used to access the PPDB can
be found in Section IX.3.4 [Hyperlink].

Access Notes

See Section IX.5.1 [Hyperlink] for general access information.

Routines RPPDCO and RPDHSH must be called before data can be read from
the PPDB.  These routines read control information from the PPDB into
common blocks used by the read/write routines.

Routines WPPDCO and WPDHSH must be called after data has been written
to the PPDB.  These routines write control information from common
blocks used by the read/write routines to the PPDB.  WPPDCO must be
called if any control information has changed including adding a new
day of daily data.  WPDHSH must be called only if stations have been
added, changed or deleted.

If data has been written to the PPDB all files must be closed by
calling routine UCLOSL.

Examples

The following example shows how to:
o write and read PP24 data value for a station
o write and read STG data value for a station

      PROGRAM MAIN
C
      CHARACTER*4 TYPE,UNITS,XTIME
      CHARACTER*8 STAID
C
C  WPD1S AND RPD1S VARIABLES
      PARAMETER (NDTYPE=1)
      CHARACTER*4 DTYPE(NDTYPE)
      PARAMETER (NDTYPR=1)
      CHARACTER*4 DTYPR(NDTYPR)
      PARAMETER (LPNTRS=10000)
      PARAMETER (LIDATA=5000)
      INTEGER*2 MSNG,IPNTRS(LPNTRS),IDATA(LIDATA)
C
C  WPDRRS AND RPDRRS VARIABLES
      PARAMETER (NRTYPE=1)
      CHARACTER*4 RTYPES(NRTYPE),RUNITS(NRTYPE)
      DIMENSION NVLPOB(NRTYPE),NUMOBS(NRTYPE)

http://www.nws.noaa.gov/ohd/hrl/nwsrfs/users_manual/part9/_pdf/942_intro.pdf
http://www.nws.noaa.gov/ohd/hrl/nwsrfs/users_manual/part9/_pdf/934_intro.pdf
http://www.nws.noaa.gov/ohd/hrl/nwsrfs/users_manual/part9/_pdf/951genl.pdf

      DIMENSION IWRITE(NRTYPE),LSTHR(NRTYPE)
      PARAMETER (NOBS=2)
      DIMENSION OBS(NOBS)
      PARAMETER (NMIN=1)
      DIMENSION MIN(NMIN)
      PARAMETER (LWKBUF=500)
      DIMENSION WKBUF(LWKBUF)
C
      INCLUDE 'upagex'
      INCLUDE 'udebug'
C
      INCLUDE 'pdbcommon/pdenqx'
C
C  COMMON BLOCKS HDFLTS AND FCTIME ARE NOT FROM INCLUDES BECAUSE
C  VARIABLES LOCAL AND NLSTZ ARE IN BOTH
C
C                                           *--> FROM HCLCOMMON.HDFLTS
C  HYDROLOGIC COMMAND LANGUAGE USER DEFAULTS
      COMMON /HDFLTS/ TIME(3),HNAMRF(2),METRIC,ZOFF,CLKZON,INTDFL,
     *                XOPEN(5),TDATES(7),LOCAL,NLSTZ,NHOPDB,NHOCAL
      INTEGER TIME,HNAMRF,ZOFF,CLKZON,XOPEN,TDATES
C                                           *--> FROM COMMON.FCTIME
      COMMON /FCTIME/ IDARUN,IHRRUN,LDARUN,LHRRUN,LDACPD,LHRCPD,NOW(5),
     *   LOCALF,NOUTZ,NOUTDS,NLSTZF,IDA,IHR,LDA,LHR,IDADAT
C
C
C  SET INPUT/OUTPUT UNIT NUMBERS
      LP=6
      ICD=5
C
C  SET DEBUG OUTPUT UNIT NUMBER AND DEBUG AND TRACE INDICATORS
      IOGDB=6
      IPDTR=0
      IPDDB=0
C
C  OPEN INPUT/OUTPUT FILES
      CALL UPRIMO
C
C  GET USER NAME
      CALL HGTUSR (PUSRID,IERR)
C
C  PRINT USER IDENTIFIER
      WRITE (LP,*) 'USER IDENTIFIER = ',PUSRID     
C
C  SET TIMING VARIABLES
      LOCAL=7
      NLSTZ=-5
      LOCALF=LOCAL
      NLSTZF=-5
      NHOPDB=12
C
C  READ DATA BASE CONTROL RECORDS
      CALL RPPDCO (IERR)
      WRITE (LP,*)
      WRITE (LP,150) 'RPPPCO',IERR
      IF (IERR.GT.0) GO TO 130
C
C  READ DATA BASE INTEGER AND CHARACTER HASH RECORDS
      IKEY=0
      CALL RPDHSH (IKEY,IERR)
      WRITE (LP,*)
      WRITE (LP,150) 'RPDHSH',IERR
      IF (IERR.GT.0) GO TO 130
C
C -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -
C
      TYPE='PP24'
      WRITE (LP,*)

      WRITE (LP,140) TYPE
C
C  GET JULIAN DATE OF FIRST AND LAST DATE OF DATA IN FILE
      DTYPE(NDTYPE)=TYPE
      CALL RPDDTE (DTYPE,JLSTRT,JLSTOP,IERR)
      WRITE (LP,150) 'RPDDTE',IERR
      IF (IERR.GT.0) GO TO 130
      WRITE (LP,*)
     *   ' JLSTRT=',JLSTRT,
     *   ' JLSTOP=',JLSTOP,
     *   ' '
C
C  CONVERT FROM JULIAN HOURS TO MONTH, DAY, YEAR AND HOUR
      NHOUR=0
      XTIME='EST'
      CALL MDYH2 (JLSTRT,NHOUR,NBMO,NBDA,NBYR,NBHR,NTZC,IDLS,XTIME)
      CALL MDYH2 (JLSTOP,NHOUR,NEMO,NEDA,NEYR,NEHR,NTZC,IDLS,XTIME)
      WRITE (LP,160) NBMO,NBDA,NBYR,NBHR,XTIME,NEMO,NEDA,NEYR,NEHR,
     *    XTIME
C
C  CHECK IF NO DATA IN DATA BASE
      NHOUR=0
      IF (JLSTRT.EQ.0) THEN
         CALL UDATEI (NMO,NDA,NYR,NHRMIN,NSEC,JULDAT,ISTAT)
         NHR=NHRMIN/100
         CALL JULDA2 (JLSTRT,NHOUR,NMO,NDA,NYR,NHR,NTZC,IDLS,XTIME)
         WRITE (LP,'(1X,A,I6)') 'JLSTRT SET TO : ',JLSTRT
         CALL MDYH2 (JLSTRT,NHOUR,NBMO,NBDA,NBYR,NBHR,NTZC,IDLS,XTIME)
         WRITE (LP,170) NBMO,NBDA,NBYR,NBHR,XTIME
         JLSTOP=JLSTRT
         NHOUR=12
         ENDIF
C
      JSTRT=JLSTRT
      IF (NHOUR.GE.12) JSTRT=JSTRT+1
      JSTOP=JLSTOP
      IF (NHOUR.GE.12) JSTOP=JSTOP+1
      JHOUR=JSTRT*24+NHOUR
      WRITE (LP,*)
     *   ' JSTRT=',JSTRT,
     *   ' JSTOP=',JSTOP,
     *   ' JHOUR=',JHOUR,
     *   ' '
C
      IENDIN=0
C
      STAID=' '
      WRITE (LP,*) 'ENTER NAME OF ',TYPE,' STATION OR <RETURN>'
      READ (ICD,'(A8)',END=10) STAID
      GO TO 20
10    IENDIN=1
20    IF (STAID.EQ.' ') STAID='SMTSE'
C
C  WRITE PP24 VALUE
      IDTYPE=0
      UNITS='IN'
      LDATA=1
      DATA=9.99
      IREV=1
      WRITE (LP,*)
     *   ' JHOUR=',JHOUR,
     *   ' '
      CALL WPD1S (STAID,IDTYPE,NDTYPE,DTYPE,UNITS,JHOUR,JHOUR,
     *   LDATA,DATA,IWRITE,IREV,IERR)
      WRITE (LP,180) 'WPD1S',STAID,IERR,IWRITE
C
C  READ DATA FOR FIRST JULIAN DAY FOR ALL STATIONS
      IRETRN=1

      LENDAT=1
      WRITE (LP,*)
     *   ' JSTRT=',JSTRT,
     *   ' '
      CALL RPDDLY (DTYPE,JSTRT,IRETRN,LPNTRS,IPNTRS,LPFILL,
     *   LIDATA,IDATA,LDFILL,NUMSTA,MSNG,LENDAT,IDATES,IERR)
      WRITE (LP,190) 'RPDDLY',IERR,DTYPE
      IF (IERR.EQ.0) THEN
         WRITE (LP,*)
     *      ' LPFILL=',LPFILL,
     *      ' LDFILL=',LDFILL,
     *      ' NUMSTA=',NUMSTA,
     *      ' NMSG=',MSNG,
     *      ' '
         IF (LPFILL.GT.0) THEN
            WRITE (LP,200)
            WRITE (LP,210) (IPNTRS(I),I=1,LPFILL)
            ENDIF
         IF (LDFILL.GT.0) THEN
            WRITE (LP,220)
            WRITE (LP,210) (IDATA(I),I=1,LDFILL)
            NSLOT=5
            DO 30 I=1,LDFILL
               JPNTRS=(NSLOT*(I-1))+1
               IF (IPNTRS(JPNTRS).EQ.0) THEN
                  WRITE (LP,225) JPNTRS
                  GO TO 30
                  ENDIF
               IF (IDATA(I).EQ.MSNG) THEN
                  WRITE (LP,230) I
                  GO TO 30
                  ENDIF
               IVAL=IDATA(I)
               IPP24=((IVAL/10)+3000)
               IHOUR=3*(((IPP24-3000)*10)+IABS(IVAL))
               PP24=IPP24/100.
               WRITE (LP,240) I,DTYPE,PP24,IHOUR
30             CONTINUE
            ENDIF
         ENDIF
C
C  READ DATA FOR FIRST JULIAN DAY FOR ONE STATION
      WRITE (LP,*)
     *   ' JSTRT=',JSTRT,
     *   ' JSTOP=',JSTOP,
     *   ' '
      CALL RPD1S (STAID,IDTYPE,NDTYPE,DTYPE,JSTRT,JSTOP,
     *   NDTYPR,DTYPR,MTYPES,LIDATA,IDATA,LDFILL,IDELT,NVPDT,MSNG,
     *   IERR)
      WRITE (LP,250) 'RPD1S',STAID,IERR,LDFILL
      IF (IERR.EQ.0) THEN
         WRITE (LP,*)
     *      ' LDFILL=',LDFILL,
     *      ' NMSG=',MSNG,
     *      ' '
         IF (LDFILL.GT.0) THEN
            WRITE (LP,220)
            WRITE (LP,210) (IDATA(I),I=1,LDFILL)
            DO 40 I=1,LDFILL
               IF (IDATA(I).EQ.MSNG) THEN
                  WRITE (LP,230) I
                  GO TO 40
                  ENDIF
               IVAL=IDATA(I)
               IPP24=((IVAL/10)+3000)
               IHOUR=3*(((IPP24-3000)*10)+IABS(IVAL))
               PP24=IPP24/100.
               WRITE (LP,240) I,DTYPE,PP24,IHOUR

40             CONTINUE
            ENDIF
         ENDIF
C
C -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -
C
      TYPE='STG'
      WRITE (LP,*)
      WRITE (LP,140) TYPE
C
      STAID=' '
      IF (IENDIN.EQ.0) THEN
         WRITE (LP,*) 'ENTER NAME OF ',TYPE,' STATION OR <RETURN>'
         READ (ICD,'(A8)',END=50) STAID
         GO TO 60
         ENDIF
50    IENDIN=1
60    IF (STAID.EQ.' ') STAID='FDKSE'
C
C  WRITE RRS DATA VALUE
      IDTYPE=0
      RTYPES(NRTYPE)=TYPE
      NVLPOB(NRTYPE)=2
      RUNITS(NRTYPE)='FT'
      NUMOBS(NRTYPE)=1
      CALL UMEMOV (JHOUR,OBS(1),1)
      OBS(2)=999.9
      IFUT=0
      IREV=1
      WRITE (LP,*)
     *   ' JHOUR=',JHOUR,
     *   ' '
      CALL WPDRRS (STAID,IDTYPE,NRTYPE,RTYPES,NVLPOB,RUNITS,NUMOBS,
     *   NOBS,OBS,NMIN,MIN,LWKBUF,WKBUF,IWRITE,IFUT,LSTHR,IREV,IERR)
      WRITE (LP,180) 'WPDRRS',STAID,IERR,IWRITE
C
C  READ DATA FOR FIRST JULIAN DAY FOR ONE STATION
      WRITE (LP,*)
     *   ' JHOUR=',JHOUR,
     *   ' '
      CALL RPDRRS (STAID,IDTYPE,RTYPES(1),NVLPOB(1),JHOUR,JHOUR,
     *   NOBS,OPS,NUMOBS,NMIN,MIN,LWKBUF,WKBUF,LSTHR,IERR)
      WRITE (LP,270) 'RPDRRS',STAID,IERR,LSTHR
      IF (IERR.EQ.0) THEN
         CALL UMEMOV (OBS(1),JHOUR,1)
         WRITE (LP,'(A,I6,A,F10.2,A,I6,A)')
     *     ' JHOUR=',JHOUR,
     *     ' OBS(2)=',OBS(2),
     *     ' LSTHR=',LSTHR,
     *     ' '
         ENDIF
C
C -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -
C
C  UPDATE DATA BASE CONTROL RECORDS
130   CALL WPPDCO (IERR)
      WRITE (LP,150) 'WPPDCO',IERR
C
C  CLOSE FILES
      CALL UCLOSL
C
      WRITE (LP,300)
      STOP
C
C -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -  -
C
140   FORMAT (' TYPE=',A)
150   FORMAT (' ',A,' CALLED : IERR=',I2)

160   FORMAT (' ',
     *   'DATE OF FIRST DATA VALUE = ',
     *      I2.2,'/',I2.2,'/',I4.4,'-',I2.2,A,
     *   3X,
     *   'DATE OF LAST DATA VALUE = ',
     *      I2.2,'/',I2.2,'/',I4.4,'-',I2.2,A,
     *   3X)
170   FORMAT (' ',
     *   'DATE OF FIRST DATA VALUE = ',
     *      I2.2,'/',I2.2,'/',I4.4,'-',I2.2,A,
     *   3X)
180   FORMAT (' ',A,' CALLED : STAID=',A,3X,'IERR=',I2,3X,
     *   'IWRITE=',I3)
190   FORMAT (' ',A,' CALLED : IERR=',I2,3X,'TYPE=',A)
200   FORMAT (' IPNTRS ARRAY:')
210   FORMAT (10(1X,I6,1X))
220   FORMAT (' IDATA ARRAY:')
225   FORMAT (' IPNTRS ARRAY POSITION ',I4,' IS MARKED DELETED')
230   FORMAT (' IDATA ARRAY POSITION ',I4,' IS SET TO MISSING')
240   FORMAT (' IDATA ARRAY POSITION ',I4,' : ',A,'=',F5.2,3X,
     *   'IHOUR=',I2)
250   FORMAT (' ',A,' CALLED : STAID=',A,3X,'IERR=',I2,3X,
     *   'LDFILL=',I5)
260   FORMAT (' IDATA ARRAY POSITION ',I4,' : ',A,'=',F5.2,3X,
     *   'IHOUR=',I2,3X,'XLAT=',F7.2,3X,'XLON=',F7.2)
270   FORMAT (' ',A,' CALLED : STAID=',A,3X,'IERR=',I2,3X,
     *   'LSTHR=',I6)
280   FORMAT (' ',A,' CALLED : IERR=',I2,3X,'TYPE=',A,3X,
     *   'MAXPST=',I5)
290   FORMAT (' ',A,' CALLED : IERR=',I2,3X,'TYPE=',A,3X,
     *   'MAXGRD=',I5)
300   FORMAT ('0- PROCESSING COMPLETED -')
C
      END

